4. Решение линейных и нелинейных уравнений и систем

4.1 Символьные вычисления

Для аналитического решения линейных и нелинейных уравнений и систем служит команда solve(eqn,per), где eqn - множество уравнений, а per - множество переменных. Напомним, что множество - совокупность разделенных запятыми объектов, взятая в фигурные скобки. Если переменные не определены, то Maple V найдет решение для всех неизвестных, упомянутых во множестве уравнений.

Например, рассмотрим систему из трех линейных уравнений.

```
> eqs1:={x+2*y+3*z=6,5*x+5*y+4*z=1,3*y+4*z=1};
```

$$eqs1 := \{x + 2\,y + 3\,z = 6,\ 5\,x + 5\,y + 4\,z = 1,\ 3\,y + 4\,z = 1\}$$

```
> a1:=solve(eqs1, {x,y,z});
```

$$a1 := \{x = \frac{42}{13}, z = \frac{82}{13}, y = \frac{-105}{13}\}$$

Мы видим, что результатом решения является также множество, каждый элемент которого доступен обычным образом. Например, найдем приближенное решение **y**, с точностью до четвертого знака.

```
> evalf(subs(a1,y),4);
```

$$-8.077$$

В следующем примере мы увидим, что результатом выполнения могут быть одно или несколько решений.

Пример 1 (Maple V выдает одно решение):

```
> eqs5:={2*x*y=1,x+z=0,2*x-3*z=2};
```

$$eqs5 := \{2\,x\,y = 1,\ x + z = 0,\ 2\,x - 3\,z = 2\}$$

```
> solve(eqs5,{x,y,z});
```

$$\{x = \frac{2}{5}, y = \frac{5}{4}, z = \frac{-2}{5}\}$$

Проверим методом подстановки правильность решения системы. Для этого воспользуемся командой subs. Здесь мы используем принятые обозначения “– последнее и “”– предпоследнее.

```
> subs(","");
```

$$\{1 = 1, 2 = 2, 0 = 0\}$$

Пример 2 (решений много):

```
>eqs6:={x+y+z=6,x^2+y^2+z^2=14,x^3+y^3+z^3=36};
```

$$eqs6 := \{x + y + z = 6, x^2 + y^2 + z^2 = 14, x^3 + y^3 + z^3 = 36\}$$

```
> solve(eqs6,{x,y,z});
```

{y = 1, z = 2, x = 3}, {y = 1, z = 3, x = 2}, {y = 2, z = 1, x = 3}, {z = 3, y = 2, x = 1}, {z = 1, y = 3, x = 2}, {z = 2, y = 3, x = 1}

Рассмотрим следующий пример. Найдем решение нелинейного уравнения ln(x) – 2x(ln(x) – 1)=0:

```
> eqn3:=ln(x)-2*x*(ln(x) – 1);
```

$$eqn3 := \ln(x) - 2\,x\,(\ln(x) - 1)$$

Следует отметить, что выражение eqn3 мы не приравнивали к нулю, т. к. это происходит по умолчанию в команде solve.

```
> a3:=solve(eqn3,x);
```

$$a3 := \mathbf{e}^{\mathrm{RootOf}(-\_Z + 2\,\_Z\,\mathbf{e}^{-\_Z} - 2\,\mathbf{e}^{-\_Z})}$$

Иногда Maple V не удается найти ответа в символьном виде, или система может выдавать его в неявной форме (как в предыдущем примере). В этом случае можно попробовать найти решение в численном виде.

4.2 Численные вычисления

Для решения уравнений численными методами служит команда fsolve:

```
> a3:=fsolve(eqn3,x);
```

$$a3 := 3.258475747$$

В качестве следующего примера попробуем найти корни полинома пятой степени.

```
> poly := 205*x^5 + 140*x^4 + 70*x^2 + 5*x:
```

Для обыкновенного уравнения команда fsolve вычисляет как минимум одиночный реальный корень. Для полинома она вычислит все реальные (несложные) корни, хотя исключительно плохо обусловленные полиномы могут заставлять fsolve пропускать некоторые корни.

Для определения режима вычисления корней используют следующие опции:

complex – находит все комплексные корни.

fulldigits – эта опция ограничивает команду fsolve от выполнения промежуточных вычислений чисел высоких порядков. С этой опцией fsolve может решить плохо обусловленные задачи, но решение займет несколько большее количество времени.

maxsols=n – находит n наименьших корней. Эта опция применяется только для полинома, у которого больше чем один корень.

interval – определяется так: a .. b или x = a .. b или {x=a .. b, y=c ..d, ...}. Поиск корней ведется только в данном интервале.

Для заданного выше полинома poly найдем три наименьших корня.

```
> fsolve( poly, x, maxsols=3 );
```

$$0, -1., -.07079255936$$

А теперь найдем все корни полинома poly на интервале от -0.5 до 0.

```
> fsolve(poly, x,-0.5..0);
```

$$0, -.07079255936$$

4.3 Целочисленные вычисления

Процедура isolve (eqn,vars) решает уравнения, в которых присутствуют только целые числа. Она находит все корни уравнения.

Необязательный второй параметр vars используется, чтобы задать имена глобальным переменным, которые имеют целочисленные значения и используются в решении. Если введен только один параметр (eqn), глобальные переменные имеют имена _N1, _N2, и т. д.. Например:

```
> isolve(2*x-5*y=8);
```

$$\{x = 4 + 5\ \_N1, y = 2\ \_N1\}$$

```
> isolve(x^2=3);
```

Isolve возвращает значение NULL, если не имеется никаких целочисленных решений или Maple неспособен найти решения.

4.4 Рекуррентные выражения

Для рекуррентных выражений используется команда rsolve.

```
> rsolve(f(n) = - 3*f(n - 1) - 2*f(n - 2),
f(k));
```

$$(2\,\mathrm{f}(0)+\mathrm{f}(1))\,(-1)^k+(-\mathrm{f}(0)-\mathrm{f}(1))\,(-2)^k$$

```
> rsolve(f(n) = 3*f(n/2) + 5*n, f(n));
```

$$\mathrm{f}(1)\,n^{\left(\frac{\ln(3)}{\ln(2)}\right)}+n^{\left(\frac{\ln(3)}{\ln(2)}\right)}\left(-15\left(\frac{2}{3}\right)^{\left(\frac{\ln(n)}{\ln(2)}+1\right)}+10\right)$$

```
> rsolve(t(b*n) = a*t(n) + n, t(m));
```

$$\mathrm{t}(1)\,m^{\left(\frac{\ln(a)}{\ln(b)}\right)}+m^{\left(\frac{\ln(a)}{\ln(b)}\right)}\left(-\frac{\left(\frac{b}{a}\right)^{\left(\frac{\ln(m)}{\ln(b)}+1\right)}a}{b\,(-b+a)}+\frac{1}{-b+a}\right)$$

4.5 Решение систем линейных уравнений

Для решения систем линейных уравнений высоких порядков можно использовать команду **`linsolve(A,B)`**, где **A[n*m]** - матрица коэффициентов системы уравнений из **m** - уравнений и **n** - неизвестных; **B** - матрица (вектор) свободных членов системы уравнений; **X -** матрица (вектор) неизвестныых параметров. Действие команды продемонстрируем на примере решения системы **eqs1,** состоящей из трех линейных уравнений:

```
> with(linalg):
```

Warning, new definition for norm
Warning, new definition for trace

```
> A:=matrix([[1,2,3],[5,5,4],[0,3,4]]);
```

$$A := \begin{bmatrix} 1 & 2 & 3 \\ 5 & 5 & 4 \\ 0 & 3 & 4 \end{bmatrix}$$

```
> B:=vector([6,1,1]);
```

$$B := [6, 1, 1]$$

```
> X:=linsolve(A,B);
```

$$X := \left[\frac{42}{13}, \frac{-105}{13}, \frac{82}{13} \right]$$

Как видим, результат тот же, что и при решении системы с помощью команды solve (в начеле главы). Выигрыш во времени при решении системы третьего порядка не ощутим. Однако, решение систем высоких порядков (более 20-го) командами solve, fsolve может занимать некоторое время. Уменьшить его поможет вышеописанный метод. При этом, в библиотеке linalg существует возможность приведения матрицы **A** к различным специальным формам, что еще может увеличить скорость поиска решения. Подробнее – смотрите библиотеку linalg.

4.6 Решение неравенств

Решить неравенства поможет уже известная команда **solve.** Использование этой команды аналогично как и при решении уравнений:

```
> solve( x^2+6*x+9>5, x);
```

$$\mathrm{RealRange}(-\infty, \mathrm{Open}(-3 - \sqrt{5})), \mathrm{RealRange}(\mathrm{Open}(-3 + \sqrt{5}), \infty)$$

В качестве результата Maple выдал два интервала, причем выражение RealRange - указывает на вещественную часть ответа, а выражение Open - на открытость границ интервала.

С помощью команды **solve** можно также решить как систему неравенств, так и систему, в которую входят и равенства и неравенства:

```
> solve( {x^2*y^3=0, x+y=2, x<>0} );
```

$$\{x = 2, y = 0\}, \{x = 2, y = 0\}, \{x = 2, y = 0\}$$